

# Count Words in Excel

Use TRIM, LEN, SUBSTITUTE and SUMPRODUCT in Excel to count words. First, let's count the total number of words in a cell or range of cells.

1. The TRIM function below returns a string with only regular spaces.

B1		✕ ✓ <i>fx</i>		=TRIM(A1)				
	A	B	C	D	E	F	G	H
1	How are you?	How are you?						
2								

2. To get the length of this string, add the LEN function.

B1		✕ ✓ <i>fx</i>		=LEN(TRIM(A1))				
	A	B	C	D	E	F	G	H
1	How are you?	12						
2								

3. The SUBSTITUTE function below returns a string without spaces.

B1		✕ ✓ <i>fx</i>		=SUBSTITUTE(A1," ","")				
	A	B	C	D	E	F	G	H
1	How are you?	Howareyou?						
2								

4. To get the length of this string, add the LEN function.

B1		X ✓ fx		=LEN(SUBSTITUTE(A1," ",""))				
	A	B	C	D	E	F	G	H
1	How are you?	10						
2								

5. To count the total number of words in cell A1, subtract the length of the string without spaces (formula from step 4) from the length of the string with only regular spaces (formula from step 2) and add 1.

B1		X ✓ fx		=LEN(TRIM(A1))-LEN(SUBSTITUTE(A1," ",""))+1				
	A	B	C	D	E	F	G	H
1	How are you?	3						
2								

Conclusion: to count the total number of words in a cell, simply count the number of spaces and add 1 to this result. 1 space means 2 words, 2 spaces means 3 words, etc.

6. To count the total number of words in the range A1:A2, add the SUMPRODUCT function and replace A1 with A1:A2.

B1		X ✓ fx		=SUMPRODUCT(LEN(TRIM(A1:A2))-LEN(SUBSTITUTE(A1:A2," ",""))+1)				
	A	B	C	D	E	F	G	H
1	How are you?	5						
2	Great, thanks!							
3								

Note: visit our page about the [SUMPRODUCT function](#) to learn more about this Excel function.

Finally, let's count how many times a specific word occurs in a cell or range of cells. This trick is pretty cool.

7. The LEN function below returns the length of the original text in cell A1.

C1		✕ ✓ <i>fx</i>		=LEN(A1)			
	A	B	C	D	E	F	G
1	dog, cat, dog, dog, horse, dog	dog	30				
2							

8. The SUBSTITUTE function below returns the string without the word "dog".

A2		✕ ✓ <i>fx</i>		=SUBSTITUTE(A1,B1,"")			
	A	B	C	D	E	F	G
1	dog, cat, dog, dog, horse, dog	dog					
2	, cat, , , horse,						

3

9. The LEN function below returns the length of this string.

A2		X ✓ fx		=LEN(SUBSTITUTE(A1,B1,""))			
	A	B	C	D	E	F	G
1	dog, cat, dog, dog, horse, dog	dog					
2	18						
3							

10. Subtract the length of the string without the word "dog" (formula from step 9) from the length of the original text in cell A1 (formula from step 7).

C1		X ✓ fx		=LEN(A1)-LEN(SUBSTITUTE(A1,B1,""))			
	A	B	C	D	E	F	G
1	dog, cat, dog, dog, horse, dog	dog	12				
2							

Conclusion: after removing the word "dog" from the original text, 12 characters are missing.

11. We know the length of the word "dog" (3), so the word "dog" occurs  $12 / 3 = 4$  times in cell A1.

C1		X ✓ fx		=(LEN(A1)-LEN(SUBSTITUTE(A1,B1,"")))/LEN(B1)			
	A	B	C	D	E	F	G
1	dog, cat, dog, dog, horse, dog	dog	4				
2							

12. To count how many times the word "dog" occurs in the range A1:A2, add the SUMPRODUCT function and replace A1 with A1:A2.

C1		X ✓ fx		=SUMPRODUCT((LEN(A1:A2)-LEN(SUBSTITUTE(A1:A2,B1,"")))/LEN(B1))			
	A	B	C	D	E	F	G
1	dog, cat, dog, dog, horse, dog	dog	6				
2	dog, pig, dog						
3							

13. Use the COUNTIF function in Excel to count the number of cells that contain a specific word.

A9		X ✓ fx		=COUNTIF(A1:A7,"*star*")					
	A	B	C	D	E	F	G	H	I
1	star								
2	moon								
3	stars								

4		9							
5	star								
6	star								
7	star12								
8									
9		5							
10									

Note: an asterisk (\*) matches a series of zero or more characters. Visit our page about the [COUNTIF function](#) to learn more about this great Excel function.



## Chapter

### ■ [Text Functions](#)

## Learn more, it's easy

### ■ [Separate Strings](#)

### ■ [Count Words](#)

### ■ [Text to Columns](#)

### ■ [Find](#)

### ■ [Search](#)

### ■ [Change Case](#)

### ■ [Remove Spaces](#)

### ■ [Compare Text](#)

### ■ [Substitute vs Replace](#)

### ■ [Text](#)

### ■ [Concatenate](#)

### ■ [Substring](#)

## ↓ Download Excel File

### ■ [count-words.xlsx](#)

---

## Next Chapter

- [Lookup & Reference Functions](#)
- 

## Follow Excel Easy



## Become an Excel Pro

- [1. Introduction](#)
  - [2. Basics](#)
  - [3. Functions](#)
  - [4. Data Analysis](#)
  - [5. VBA](#)
- 

Count Words • © 2010-2026

Start with Excel in 2026: Range • Formulas and Functions • Ribbon • Cell References • Sort • Filter